
Perl stands for Practical Extraction and Report Language. It was designed by Larry Wall.

Perl Help

Command line help:

perl -v



checks which perl version is installed.

perldoc -f function_name
prints the documentation of a built-in function

perldoc -q regexp


Search the text of questions (not answers) in perlfaq[1-9]

perldoc module_name

prints the documentation (synopsis, usage etc.) of a module

Examples: 

perldoc –f print

perldoc –q match

perldoc IO::File

Help is also available on the web:

http://search.cpan.org/perldoc?IO::File

Running Perl programs

To run a Perl program from the Unix command line:

    perl progname.pl

Alternatively, put this as the first line of your script:

    #!  /usr/bin/perl

... and run the script as /path/to/script.pl. Of course, it'll need to be executable first, so chmod 755 script.pl (under Unix).

To run perl code from the command line, use the -e option:

    perl -e '$x=3; print 2 * $x, "\n";'

Basic syntax overview

A Perl script or program consists of one or more statements. These statements are simply written in the script in a straightforward fashion. There is no need to have a main() function or anything of that kind.

Perl statements end in a semi-colon:

    print "Hello, world";

Comments start with a hash symbol and run to the end of the line


# This is a single-line comment


=a


This is a multi-line comment


=cut

White space is irrelevant:

    print

        "Hello, world"

        ;

... except inside quoted strings:

    # this would print with a line break in the middle

    print "Hello

    world";

Double quotes or single quotes may be used around literal strings:

    print "Hello, world";

    print 'Hello, world';

However, only double quotes "interpolate" variables and special characters such as newlines (\n):

    print "Hello, $name\n";     # works fine

    print 'Hello, $name\n';     # prints $name\n literally

Numbers don't need quotes around them:

    print 42;

You can use parentheses for functions' arguments or omit them according to your personal taste. They are only required occasionally to clarify issues of precedence.

    print("Hello, world\n");

    print "Hello, world\n";

Constants

use constant PI => 3.14159;  # declare a constant symbol

Perl variable types

Perl has three main variable types: scalars, arrays, and hashes.

Scalars

A scalar represents a single value:

    my $animal = "camel";

    my $answer = 42;

Scalar values can be strings, integers or floating point numbers, and Perl will automatically convert between them as required. There is no need to pre-declare your variable types, but you have to declare them using the my keyword the first time you use them. (This is one of the requirements of use strict;.)

Scalar values can be used in various ways:

    print $animal;

    print "The animal is $animal\n";

    print "The square of $answer is ", $answer * $answer, "\n";

    print "${animal}back";
# interprets $animal as the scalar

warnings and strict mode

Perl by default is very forgiving. In order to make it more robust it is recommended to start every program with the following lines:

    #!/usr/bin/perl

    use strict;

    use warnings;

The two additional lines request from perl to catch various common problems in your code. They check different things so you need both. A potential problem caught by use strict; will cause your code to stop immediately when it is encountered, while use warnings; will merely give a warning (like the command-line switch -w) and let your code run. 

Variable scoping

Throughout the previous section all the examples have used the syntax:

    my $var = "value";

The my is actually not required; you could just use:

    $var = "value";

However, the above usage will create global variables throughout your program, which is bad programming practice. 

my limits the scope of the variable to the block (i.e. a bunch of statements surrounded by curly-braces) in which it is defined.

    my $x = "foo";

    if (1) {

        my $y = "bar";

        print $x;           # prints "foo"

        print $y;           # prints "bar"

    }

    print $x;               # prints "foo"

    print $y;               # prints nothing; $y has fallen out of scope

                            # with use strict; a compile-time error is raised

Operators

Arithmetic

$a = 3 - 4;

# Subtracts 4 from 3. Stores in $a

$a = 3 + 4;

# Adds 3 and 4. Stores in $a

$a = 3 * 4;

# Multiply 3 and 4

$a = 3 / 4;

# Divide 3 by 4

$a = 3 ** 4;

# three to the power of four

$a = 3 % 4;

# Remainder of 3 divided by 4

++$x;


# Increment $x. Then return its value

$x++;


# Return $x. Then increment its value

--$x;


# Decrement $x. Then return its value

$x--;


# Return $x. Then decrement its value

$b = ($a > 1) ? "yes" : "no";
# The ternary operator. Equivalent to..






# if ($a > 1) {$b ="yes"} else {$b = "no"};

Assigning values

$x = $y;

# Assign $y to $x; This statement makes a copy of $y and then assigns it to $x. 




# The next time you change $y it will not change $x.

$x += 5;

# same as $x = $x + 5

$x -= 5;

# same as $x = $x - 5

String operators

.   string concatenation

$x = $y . $z;

# Concatenate $y and $z

$x .= $y;

# Append $y onto $x. same as $x = $x . $y

Example:

$a = 'hello';

$b = 'world';

print $a . ' great '. $b . "\n";
# dot operator

print $a , ' great ', $b , "\n";
# list

print "$a great $b\n";

# interpolation

x   string multiplication

$x = $y x $z;



# $y repeated $z times

print "-" x $screen_width, "\n";
# prints ------ across the screen

$fill x= 80;            

# Make string $fill into 80 repeats of itself

Numeric comparison

==  equality

!=  inequality

<   less than

>   greater than

<=  less than or equal

>=  greater than or equal

<=> $a <=> $b produces 0 if equal, 1 if $a greater, -1 if $b greater

String comparison

eq  equality

ne  inequality

lt  less than

gt  greater than

le  less than or equal

ge  greater than or equal

cmp compare (see <=>)

Why do we have separate numeric and string comparisons? Because we don't have special variable types, and Perl needs to know whether to sort numerically (where 99 is less than 100) or alphabetically (where 100 comes before 99).

Boolean logic

&&  and

||  or

!   not

$val ||= "2";

# Set $val to 2 if it isn't already set

Conditional statements

if

    if ( condition ) { ... }

    elsif ( other condition ) { ... }

    else { ... }

There's also a negated version of it:

    unless ( condition ) { ... }

This is provided as a more readable version of if (!condition).

The braces are required even if the block has a single line. 

However, there is a way to write one-line conditional blocks more English like:

    print "Yow!" if $bananas;

    print "We have no bananas" unless $bananas;

while

    while ( condition ) { ... }

    until ( condition ) { ... }

# a negated version

    print "bananas\n" while 1;

# as a post-condition (loops forever)

    while ( condition ) {

       ...

       if ( condition ) { next; }

# skips to end of current loop iteration

       if ( condition ) { last; }

# skips and exits loop

       if ( condition ) { redo; }

# redo the current loop iteration from the top

    }

    LINE: while ( condition ) {

# LINE is the loop label

       next LINE if statement...

       last LINE if statement...

    }

Arrays

An array represents a list of values:

    my @animals = ("camel", "llama", "owl");

    my @animals = qw(camel llama owl);
# qw (quoted words) interprets words like single–quoted strings

    my @animals = qw/camel llama owl/;
# qw will take any punctuation character as the delimeter

    my @numbers = (23, 42, 69);

    my @mixed   = ("camel", 42, 1.23);

    my @sequence = (1..5);


# ..  range operator (creates a list of numbers)

Arrays are zero-indexed. Here's how you get at elements in an array:

    print $animals[0];

# prints "camel"

    print $animals[1];

# prints "llama"

    print $animals[2];

# prints "owl"

The printed elements start with a $ because we're getting just a single value (a scalar) from the array.
@array where Perl expects to find a scalar value ("in scalar context") will give you the number of elements in the array:

$a = @animals;


# $a is assigned the size of the array

print scalar @array;

# prints the array size

print @array;



# array items are not separated by a space
$, contains the element separator for the print command. Defaults to a null character.

print "@array";


# array items separated by a space

$" contains the array element separator. Defaults to a single blank space.

$#array tells you the index of the last element of an array (which equals the array size - 1).

($a) = @animals;


# $a is assigned the first element in the array

($a, $b) = @animals;

# $a and $b are assigned the first and second elements in the array

@array = (@animals, 'bat');
# combining an array and a scalar

@array = (@animals, @numbers);
# combining two arrays

To get multiple values from an array, i.e., an array slice:

@animals[0,1];


# gives ("camel", "llama");

@animals[0..2];


# gives ("camel", "llama", "owl");

@animals[1..$#animals];

# gives all except the first element

@array = split(/,/ , 'ant,beatle,flea');# split a scalar into an array

@array = split(// , 'ant,beatle,flea');
# split into array of single characters







# split() takes a /pattern/ - see regular expressions.

$string = join(' ', @array);

# join array items into a string scalar, space-separated

unshift(@array, 'element');

# adds an item at the beginning of the array

$element = shift(@array);


# removes the first item of the array. returns the removed item

push(@array, 'element');


# adds an item at the end of the array

$element = pop(@array);


# removes the last item of the array. returns the removed item

@reversed = reverse @numbers;

# Reverse the array

splice(@array, 2, 3);


# Removes 3 elements from array starting with $array[2]

splice(@array, 2 , 3, @list);

# Removes 3 element starting with $array[2] 







# and replaces them with the items in @list

splice(@array, 2 , 0, 'item');

# Inserts 'item' after the 2nd element of the array

@sorted = sort @animals;


# Sorts in ASCII order

@sorted = sort { $b cmp $a } @animals;
# Sorts in reversed ASCII order

@sorted = sort { $a <=> $b } @numbers;
# Sorts numerically in ascending order

@sorted = sort { $b <=> $a } @numbers;
# Sorts numerically in descending order

@sorted = sort { lc($a) cmp lc($b) } @li;
# Sorts in alphabetical order

Efficient sort by a computable character of the array, for example, by the array elements length:

(but read about the map function and about references first)

@temp = map { [ length $_, $_ ] } @strings;
# create array of array references

@temp = sort { $a->[0] <=> $b->[0] } @temp;
# sort by first element in each array reference

@sorted = map { $_->[1] } @temp; 

Loops

for ($i = 0; $i <= $max; $i++) {

   ...

}

The C style for loop is rarely needed in Perl since Perl provides the more friendly list scanning foreach loop.

Looping through an array (The two loops are equivalent):

	foreach $item (@array) {

  print $item;


}
	for $item (@array) {

  print $item;


}


Looping through an array using the special variable $_ (The two loops are equivalent):

	foreach (@array) {

   print;

}
	foreach $_ (@array) {  

   print $_;

}


$_ is a system variable. Many functions and operators will modify the contents of $_ if you do not explicitly specify a scalar variable on which they are to operate (e.g. the foreach function).

$_ acts as the default parameter for many functions when no other variable is explicitly specified (e.g. the print function).

print $list[$_] foreach 0 .. $max;

print foreach @array;

The map function evaluates a block or an expression for each element of the list provided.

Each statement on the left and the equivalent loop on the right create an array of upper case elements.

	@new_array = map (uc, @array);

@new_array = map {uc} @array;
	foreach (@array) {

   push (@new_array, uc $_);

}


Hashes

A hash represents a set of key/value pairs:

    my %fruit_color = ("apple", "red", "banana", "yellow");

You can use whitespace and the => operator to lay them out more nicely:

    my %fruit_color = (

        apple  => "red",

        banana => "yellow",

    );

To get at hash elements:

    $fruit_color{"apple"};            # "red"

    @fruit_color{"apple", "banana"};  # "red", "yellow"

You can get at lists of keys and values with keys() and values().

    my @fruits = keys %fruit_colors;

    my @colors = values %fruit_colors;

Looping through a hash:

foreach my $key (keys %fruit_color) {

  print "The value of $key is $fruit_color{$key}\n";

}

while (($key, $value) = each(%fruit_color)) {

  print "The value of $key is $value\n";

}

$fruit_color{'grape'} = "blue";

# Add a hash key-value pair

$deleted =  delete $fruit_color{'apple'};
# Delete a hash element and return its value

print exists $fruit_color{'banana'};
# Check if key exists

if ($fruit_color{'banana'} eq "") {...};
# Check if key exists

if (%fruit_color) {...};


# Check if hash is empty

%hash = map { get_a_key_for($_) => $_ } @array;
# generate %hash with the map function

Files and I/O

You can open a file for input or output using the open() function. 

open(FILEHANDLE,   "input.txt") or die "Can't open  input.txt: $!";
# open file for read

open(FILEHANDLE,  "<input.txt") or die "Can't open  input.txt: $!";
# same thing, explicitly

open(FILEHANDLE, ">output.txt") or die "Can't open output.txt: $!";
# create file for write or overwrite

open(FILEHANDLE,  ">>file.txt") or die "Can't open   file.txt: $!";
# append to existing file or create

close FILEHANDLE or warn $! ? "Error closing file: $!" : "Exit status $?";

$! is the system error message variable. It holds the current value of errno. In string context, the associated string is returned.

You can read from an open filehandle using the <> operator. In scalar context it reads a single line from the filehandle, and in list context it reads the whole file in, assigning each line to an element of the list:

    my $line  = <FILEHANDLE>;

    my @lines = <FILEHANDLE>;

Reading in the whole file at one time is called slurping. It can be useful but it may be a memory hog. Most text file processing can be done a line at a time with Perl's looping constructs.

    while (<FILEHANDLE>) {     # assigns each line in turn to the default variable $_

        print "$.: $_";

    }

print while <FILEHANDLE>;   is equivalent to   print $_ while defined($_ = <FILEHANDLE>);

$. contains the line number of the last line read from an input file.

$/ contains the current input line separator. Newline character is the default.

   $/ = "->";  Sets the input line separator to ->. It will keep reading a line until it hits ->

$\ contains the current output line separator. It is set to a null character by default, which means no output. 

     This is automatically printed after every call to print.

We've already seen how to print to standard output using print(). However, print() can also take an optional first argument specifying which filehandle to print to:

    print FILEHANDLE $record;

    print STDERR "This is your final warning.\n";

Solicit and read user input

print STDOUT "Enter a number: ";
# STDOUT is the default output stream

chop($number = <STDIN>);        
# chop removes (and returns) the last character passed to it.

chomp($number);                 
# chomp removes newline from end of string

Supplying arguments when running a perl program from the command line

@ARGV  holds the words typed after the program name.

   while (<>) { ... }

003C> processes any filename typed on the command line. If no filenames were given, the program reads from STDIN. In the list of filenames, "-" indicates STDIN and "someprogram |" indicates the output of another program.

-n adds the while (<>) loop around your program text.

   #! /usr/local/bin/perl

   while (<>) { print if /adam/; }  # prints all lines containing the string "adam"

can be written as

   #! /usr/local/bin/perl -n
   print if /adam/;

or as a one line perl code on the command line:

   perl -ne 'print if /adam/;' filename
-p is like -n but it adds a print; at the end of the loop

   perl -pe 'tr/[a-z]/[A-Z]/' filename
Working with Files and Directories

-e "file_name"  # True if file exists

-r "file_name"  # True if file readable

-w "file_name"  # True if file writable

-d "file_name"  # True if file is a directory

-f "file_name"  # True if file is a regular file

-T "file_name"  # True if file is a text file

flock(FILEHANDLE, 2);  # After opening a file, get exclusive access to it

flock(FILEHANDLE, 8);  # Release the file.

rename("../../old/file.name", "../../new/file.txt") or die;  # rename and/or move a file

unlink("../../dir/file.txt") or die;
# delete a file

opendir(DIR, "path/directory") or die;
# open a directory, with handle label DIR

@myfiles = readdir(DIR);


# assign directory content to an array

closedir(DIR);



# close directory

chdir("../../new/dir");


# change working directory

mkdir("../../new/dir", 0777) or die;
# create a new directory with 755 permissions (with umask 22)

chmod(0777, "../../dir") or die

# change file or directory permission

rmdir("../../dir") or die;

# remove a directory

Regular expressions

There are three main uses for regular expressions in Perl: matching, substitution, and translation. 

m//
m/pattern/


# returns true if pattern is found

s///
s/pattern/replacement/

# replaces the substring matched by pattern with replacement

tr///
tr/characters/replacements/
# replaces characters with the characters in replacements

$line = "Chickens in nature may live for five to eleven years";

if ($line =~ m/nature/) { ... }
# true if $line contains "nature"

if ($line =~  /nature/) { ... }
# equivalent

=~ is the binding operator, not an assignment operator. It tells Perl to match the pattern on the right against the string on the left, instead of matching it against the default special variable $_.

if (/nature/) { ... }

# true if $_ contains "nature"

$line =~ s/nature/nurture/;
# replaces the first occurrence of nature with nurture in $line

s/nature/nurture/;

# replaces the first occurrence of nature with nurture in $_

$count = $line =~ s/nature/nurture/g;
# global search: replaces all instances of nature with nurture






# returns the number of substitutions performed

$line =~ s/nature/nurture/i;
# case insensitive search

$count = $line =~ tr/a-z/A-Z/;
# convert to upper case. returns the # of translated characters

$line = "Hens of special laying breeds may produce as many as 300 eggs a year.";

$line =~ s/(\d+)/$1 * 2/e;
# evaluate replacement as code rather than as a string

$line =~ s/(\d+) # hens/2/x;
# allow space and comments in pattern

$&
contain the data that the pattern matched

$`
contain any text that preceded the matched text

$'
contain any text that came after the matched text

 Patterns syntax:

/\bChicken\b/
# \b matches word boundary. will match "Chicken" but not "Chickens"

/^Chicken/
# ^ matches pattern at start of string

/years$/

# $ matches pattern at end of string

/\d/

# matches one digit

/\d{3}

# matches exactly 3 digits

/\d{3,}

# matches 3 or more digits

/\d{4,8}/
# matches between 4 and 8 digits

/\d+/

# + {1, } matches 1 or more digits

/\d*/

# * {0, } matches 0 or more digits

/\d?/ 

# ? {0, 1} matches 0 or 1 digits

/.ive/

# . acts as a wild character. matches "live" and "five"

/.*a/

# greedy search, will match the maximum number of characters 



# before the last occurring 'a' in the string

/.*?a/

# non-greedy search, will match the minimum possible number of characters

/[aeiou]/        # matches a single character in the given set

/[^aeiou]/       # matches a single character outside the given set

/[a^A]/

# matches a, A or the caret (^)

/(egg|hen|owl)/  # matches any of the alternatives specified

/(\d\s){3}/      # three digits, each followed by a whitespace character (eg "3 4 5 ")

digit           [0-9]           \d      non-digit               \D
white space     [ \t\n\r\f]     \s      non-white space         \S
word character  [a-zA-Z_0-9]    \w      non-word character      \W 

word boundary
               \b      non-word-boundary       \B
/\w+/

# matches the first word

To match any of the special characters []()*.^-|\? precede it with a backslash:

/\\w/

# matches a backslash followed by w

$a++ while $DNA =~ /A/ig;
# counts  the number of 'A's in $DNA (/g sets pointer to end of last match)

$a = ($DNA =~ tr/Aa//);
# returns the number of 'A's in $DNA

Parentheses are used for grouping. They can also be used to capture the results of parts of the regexp match for later use.

The results end up in $1, $2 and so on.

s/(\w+)\W+(\w+)/$2 $1/

# substitute 1st and 2nd patterns found in parentheses

if ($email =~ /([^@]+)@(.+)/)  {
# break an e-mail address into parts

   print "Username: $1, Hostname: $2\n";  }

String manipulation functions

length('str');
# returns the string length

$tmp_ptr = index($string, "sub-string"); # find first occurrence of sub-string in string

$my_sub_string = substr($string, $start_position, $length);

uc('str');
# returns STR. upper case

lc('STR');
# retruns str. lower case

ucfirst('str');
# returns Str. upper case first character

lcfirst('STR');
# returns sTR. lower case first character

Subroutines

Here are some examples of how to create functions in Perl:

sub sub1 {

  print "sub1 called\n";

}

sub1();


# calling subroutine sub1

&sub1;


# an equivalent way to call subroutine sub1

sub1;


# another way to call sub1

# A subroutine can have a return value. If a subroutine doesn't have an explicit return value, 

# the default return value is the last expression evaluated.

sub sub2 {

  print "sub2 called\n";

  return 5;

}

my $var1 = sub1(); 
# call sub1 and store the return value in $var1

my $var2 = sub2();  


print "returned value of subroutine sub1: $var1\n";

print "returned value of subroutine sub2: $var2\n";

# You can submit arguments to a subroutine:

sub3("agent", 700);


# @_ is a special variable defined inside each subroutine. 

# It contains an array of all the arguments passed to the subroutine.

sub sub3 {

  print "Arguments sent to sub3: @_ \n";
# @_ contains the arguments to sub3.

  print "First argument:   $_[0] \n";
# $_[0], $_[1] are the elements in the @_  array

  print "Second argument:  $_[1] \n";

  my $var3 = shift;


# The default argument to the shift function is @_

     # this expression shifts the first item off the list of arguments and assigns it to $var3

  print "First argument: $var3 \n";

}

Subroutines can be stored in a separate file. The last line in the file should be 

1;

When you call the subroutine, this value will return true, and Perl will know that the external file was accessed successfully.

To have access to subroutines that are contained in an external file named subroutines.pl, type 

require 'subroutines.pl';

The local keyword limits the scope of variables to the block inside the subroutine and to any subroutines called by the subroutine. In comparison, my limits the scope of variables to the local block:
my $a = 1;

&sub1;

sub sub1 { my $b = 2; local $c = 3; &sub2; }

sub sub2 { print "a: $a  b: $b c: $c\n"; }

# $b is out of scope

print "a: $a  b: $b c: $c\n";


# $b, $c are out of scope

References

Each defined variable has a name and the address of a chunk of memory associated with it.

A reference is a value that holds the location of another value.

Reference Assignment


How to Dereference    (works without the { } too)                       .

$refScalar   = \$scalar;

${$refScalar}   is a scalar value. 

$refArray    = \@array;

@{$refArray}    is an array value. 

$refHash     = \%hash;

%{$refHash}     is a hash value. 

$refFunction = \&function;
&{$refFunction} is a function location. 

$refRef      = \$refScalar;
${${$refScalar} is a scalar value.

$refglob     = \*file;

Glob references are beyond the scope of this course 

Examples:

$scalar = 3;

$refS = \$scalar;

# declare a reference to a $scalar

print "$$refS\n";

# dereference $refS

@array = qw/Adenine Cytosine Thymine Guanine/;   

$refA = \@array;


# declare a reference to @array

print "$refA->[0] \n";

# get the first element of the array

print "${$refA}[0]  \n";

# dereference $refA and extract the first element

print "$$refA[0]\n";

# as above

print "@$refA \n";

# dereference $refA to get @array

my %hash = ("apple", "red", "banana", "yellow");

$refH = \%hash;


# declare a reference to %hash

print "$refH->{'apple'} \n";
# get a value by its key in the hash

sub mySub { print "Arguments sent to mySub: @_\n"; }

$refSub = \&mySub;

# declare a reference to mySub()

$refSub->('agent', 700);

# run mySub() through a reference

&$refSub('agent', 700);

# dereference $refSub to run mySub()

You can create references to unnamed data:

$pi = \3.14159;


# creates a reference to a constant value

$$pi = 4;


# an attempt to change the referent will cause a runtime error

$refA = [3, 4, 5];

# creates a reference to an anonymous, modifiable array

$refH = {"apple"=>"red", "banana"=>"yellow"}; # creates a reference to an anonymous hash

$refS = sub { print "This is an anonymous subroutine\n"; };

Why use references:

A) When you pass data to a function by value, the function creates its own copy of the data. 

When you pass data to a function by reference, the function uses the original data.

If the data consume a lot of memory, the latter method saves memory space and computation time.

sub pass_by_val {

  my $var = shift @_;

  $var = 8;

  print 'Within pass_by_val $var = ', "$var\n";   

}

sub pass_by_ref {

  my $var = shift @_;

  $$var = 8;

  print 'within pass_by_ref $$var = ', "$$var\n";

}

my $var = 5;

pass_by_val($var);




# pass-by-value

print 'Outside the function block $var = ', "$var\n";

pass_by_ref(\$var);




# pass-by-reference

print 'Outside the function block $var = ', "$var\n";

B) We can't pass more than one array to a function by value. This is because functions only see one array (the @_ array) when looking for parameters. References can be used to overcome this limitation. 

pass_by_val ( (1..5), ("A".."E") );

sub pass_by_val {

    my(@firstArray, @secondArray) = @_ ;

    print("The first array is  @firstArray.\n");

    print("The second array is @secondArray.\n");

}

pass_by_ref ( [1..5], ["A".."E"] );

sub pass_by_ref { 

    my($ref_firstArray, $ref_secondArray) = @_;

    print("The first array is  @{$ref_firstArray}.\n");

    print("The second array is @{$ref_secondArray}.\n");

}

C) The classic use of references in Perl is to circumvent the restriction that arrays and hashes may hold scalars only. 

References are scalars, so to make an array of arrays, we make an array of array references.

The ref() function can  protect program code that dereferences variables from producing errors when the wrong type of reference is used.

print ref( 10 ), "\n";

# undefined

print ref( \10), "\n";

# SCALAR

print ref([3,4,5]), "\n";

# ARRAY 

print ref({1 => "Joe"} ), "\n";
# HASH  

print ref(sub {print 1;}), "\n";
# CODE

print ref( \\10), "\n";

# REF

Arrays of Arrays

@AoA = ([8,9,1], [4,5,16], [7,8,9]);

$AoA[0][0] = 3;
# access and change the first element of the array

Print array of arrays:

for (my $i = 0; $i < @AoA; $i++) {

    for (my $j = 0; $j < @{$AoA[$i]}; $j++) {

         print $AoA[$i][$j] . " ";

    }

    print "\n";

}

Another way to print array of arrays:

for $aref ( @AoA ) {      

    print "@$aref \n";

}

Sort array of arrays in ascending order of 2nd item in array:

@sorted = sort { $a->[1] <=> $b->[1] } @AoA;

Hashes of Arrays

%insects = (

    bees => [ "honeybee", "bumblebee" ],

    beetles => [ "weevil", "firefly", "ladybug" ],

    mosquitoes => [ "anopheles", "aedes" ]

);

Records (Hashes of Hashes)

$record = {

  name => "Adam",

  age => 22,

  dependents => [ "Ben", "Ann" ]

};

# Create a structure that can hold many records - a hash of hashes:

%DB = ($record->{name} => $record);

# Looping through a hash of hashes:

while (($key, $value) = each(%DB)) {

  print "${key}'s age is $value->{age}\n";

} 

# Getting a hash element:   

print $DB{"Adam"}->{age};

Modules

A Perl module is a file that has a collection of functions designed to be used by other programs.

:: in the module name are translated into your system's directory separator.

use Directory::Module;


# imports myModule.pm at compile time

use Module qw(method1, method2);

# imports only specific methods

require Directory::Module;

# imports myModule.pm at run time

If the path is relative, then use() will attempt to search for the file in all the directories listed in @INC.

To use modules that are not installed in the standard directories, add a directory to Perl's search path @INC:
use lib '/path/to/dir';

h2xs -XA -n NewModule
 # Create a skeleton for a new module

The last line in the module must be 1; to indicate successful execution of any initialization code.
A perl module must have the file extension ".pm".

Checking if a specific module is installed from the command line:

perl -e "use CGI"

Running shell commands

Three ways to execute the sell command pwd (print working directory):

print `pwd`;

system('pwd');

use Shell "pwd"; print pwd;

Command line options

The following Perl program accepts command line options. Run it with:

  perl program_name -run -nooption -gene FGF -species worm -gpen 8 -chr 3 -weight 5.6
use Getopt::Long;

GetOptions(


"run"

=> \$run,
# boolean value, set false if not specified


"option!"
=> \$option,
# user have to specify -option or -nooption


"gene=s"

=> \$gene_name,
# =s mandatory string parameter


"species:s"
=> \$species,
# :s optional string parameter


"gpen=i"

=> \$gap_penalty,# =i mandatory integer parameter


"chr:i"

=> \$chr_number,
# :i optional integer parameter


"weight:f"
=> \$weight
# :f option float parameter

);

if (!defined $gene_name || !defined $gap_penalty)  {


usage();


exit;

}

print "run is set\n" if $run;

print "option is set\n" if $option;

print "species: $species\n" if $species;

print "gap penalty: $gap_penalty\n" if $gap_penalty;

print "gene_name: $gene_name\n" if $gene_name; 

print "chromosome number: $chr_number\n" if $chr_number;

print "weight: $weight\n" if $weight;

If an option is mandatory, perl will issue a warning only if -option is specified on the command line, but no argument is specified (or worse, it will take the next -optionb as the value, so I would set everything to optional).

Options may be abbreviated, as long as the abbreviations are unambiguous (e.g. -wei instead of -weight).

Options and other command line arguments can be mixed; the other arguments will be stored in @ARGV.

Objects

Namespace

package Test;   # packages partition the global namespace

sub tease { print "tease function called\n"; }

package main;   # "main" is the default namespace

Test::tease();  # the global function "tease" resides in the Test package

Data type

$object = {};                   # a reference to an anonymous hash

print ref($object), "\n";       # HASH

bless($object, Test);           # changes the data type of $object to "Test"

                                # and returns $object

print ref($object), "\n";       # Test

Classes

package Birds;

sub new {                       # class constructor

  my $class = shift;

  my $self = { name => undef, age => 0 }; # initialize object properties

  bless $self, $class;          # a class object is returned

}

sub DESTROY {                   # class destructor

  my $self = shift;

  printf("$self dying at %s\n", scalar localtime);

}

package main;

$bird = Birds->new();           # instantiate an object

$bird = new Birds;              # instantiate an object, another method

$bird = Birds::new('Birds');    # instantiate an object, another method

$bird = Birds'new('Birds');     # instantiate an object, yet another method

$bird->{name} = "Red Robin";

print "$bird->{name} $bird->{age} \n";

Objects are normally anonymous hashes.

Perl takes the name of the class from in front of the -> operator and adds it to the beginning of the parameter array, which is passed to the new() function.

Class destructors are called when there are no more references to the object, or when the program shuts down. No memory deallocation is needed.

package Beetles;

sub new {

  my $class = shift;

  my $name = shift;

  my $age = shift;

  my $self = { name => $name, age => $age };

  bless $self, $class;

}

package main;

$beetle = Beetles->new("Ladybug", 3);

print "$beetle->{name} $beetle->{age} \n";

Using named parameters

package Flies;

sub new {

  my $class = shift;

  my %params = @_;

  my $self = { name => $params{name}, age => $params{age} };

  bless $self, $class;

}

package main;

$fly = Flies->new(name => "Aedes", age => 2);

$fly = Flies->new("name", "Aedes", "age", 2); # as above

print "$fly->{name} $fly->{age} \n";

Class methods

package Person;

sub new {                       # class constructor

  my $class = shift;

  my $self = { name => undef, age => 0 };

  bless $self, $class;

}

sub age {                       # get and set class methods combined

  my $self = shift;

  my $old = $self->{age};

  if (@_) { $self->{age} = shift }; # set age if parameters are provided

  return $old;                      # return previous age

}

package main;

$obj = new Person;

$obj->age(5);           # set age

print $obj->age, "\n";  # get age

Inheritance

package Employee;

@ISA = (Person);

Inheritance is accomplished by placing the names of parent classes into a special array called @ISA. The elements of @ISA are searched left to right for any missing methods. In addition the UNIVERSAL class is invisibly tacked on to the end of the search list.

package main;

$obj = new Employee;

print $obj->age, "\n";

Polymorphism: methods defined in the derived class override methods defined in the parent classes.

Connecting to a MySQL database

use DBI;

$db_handle = DBI->connect("DBI:$DBI:database=$database;host=$host", $user, $password)

    or die "Couldn't connect to database: $DBI::errstr\n";

$sql = "SELECT * FROM user";

$sth = $db_handle->prepare($sql) or die "Couldn't prepare query '$sql': $DBI::errstr\n";

$sth->execute() or die "Couldn't execute query '$sql': $DBI::errstr\n";

$number_of_rows_fetched = $sth->rows;

# method a

while(@row = $sth->fetchrow_array) {

    print "User $row[0] has privileges on $row[1].\n";

}

# method b

while ($ref = $sth->fetchrow_hashref) {

    print "User $ref->{User} has privileges on $ref->{Host}.\n";

}

# method c

my($user, $host) = ('','');

$out->bind_col(1, \$user);

$out->bind_col(2, \$host);

while ($sth->fetch) {

    print "User $user has privileges on $host.\n";   

}

$sth->finish;

$db_handle->disconnect or warn "Error disconnecting: $BI:: errstr\n";

Web Programming

Directing warnings to browser for debugging purposes:

use CGI::Carp qw/fatalsToBrowser warningsToBrowser/;

Sending a cookie and indicating the Internet media type:

 print "Set-Cookie:name1=value1;name2=value2;domain=.sfu.ca; expires=Fri, 31-Dec-2000 23:59:59 GMT;\n";

 print "Content-type: text/html\n\n";

The expiration date tells the browser when to delete the cookie. If no expiration date is provided, the cookie is deleted at the end of the user session, that is, when the user quits the browser. Cookies are identified by the triple name/domain/path. To delete a cookie, sent the cookie again with an expiration date set to a past date. The domain and path tell the browser that the cookie has to be sent back to the server when requesting URLs of a given domain and path. If not specified, they default to the domain and path of the page that set the cookie. For security reasons, the cookie is accepted only if the server is a member of the domain specified by the domain string. If you set domain=.sfu.ca, the cookie will also be sent to domain search.sfu.ca.

path=/ means that the cookie will be valid throughout the entire domain.
use CGI::Cookie;

print $cgi->header( -cookie => new CGI::Cookie(-name => 'Monster',

                                         -value => $cgi->param('Monster'),

                                         -expires => '+3d',

                                         -domain => '.sfu.ca')

            );

Printing all environment variables:

foreach $key (sort keys(%ENV)) {

      print "$key = $ENV{$key}<p>";

   }

$ENV{'REMOTE_ADDR'} holds the user IP address.

$ENV{'QUERY_STRING'} is a string of variable and values sent by the GET method.

Reading all data sent by the GET or POST method:

use CGI;

$query = new CGI;

my %data = $query->Vars;

foreach my $param (keys %data) {

   print "<br> $param = $data{$param}\n";

}

$var1 = $query->param("var1");
# reading a specific variable sent by GET or POST

$var2 = $query->cookie("var2");
# reading a variable stored in a cookie

Alternative Syntax:

use CGI qw(:standard); # imports a subset of the CGI module, rather than the whole thing.

print header;

print start_html('Hello!');

print h1('Hello '. param('name') . '!');

print end_html;

Debugging: run web_page.pl from the command line

Send output via e-mail

Most Unix systems include an e-mail program called sendmail that can be used to send the output from a form via e-mail.

To find the path to this program on your Unix server, type whereis sendmail on the command line.

open (MAIL, "|/usr/lib/sendmail -t") or die "not happy\n";

print MAIL "To: sblay \nFrom: sblay\n";

print MAIL "Subject: perl is cool\n";

print MAIL "Here comes the mail content.\n";

close(MAIL);


